title: Is this a Module, Sub-Project, Library Project or something else?
body: <p>I am trying to install Android-GPU-Image in my project. <a href="https://github.com/CyberAgent/android-gpuimage" rel="nofollow noreferrer">https://github.com/CyberAgent/android-gpuimage</a></p>
  
  <p>When I download the source, it's broken down as such:</p>
  
  <p><img src="https://i.stack.imgur.com/ZgXUD.png" alt="enter image description here"></p>
  
  <p>The github project does not include install instructions (other than a dependency line that didn't work for me) so I am trying to figure out how to install a package like this.</p>
  
  <p>My question is: <strong>What is the general name of this 'android-gpuimage' folder in the context of adding it to a project? Is this a Module, Sub-Project, Library-Project or what?</strong></p>
  
  <p><em>Update</em></p>
  
  <p>Here is my gradle file</p>
  
  <pre><code>import java.util.regex.Pattern
  
  apply plugin: 'android'
  
  buildscript {
      repositories {
          mavenCentral()
      }
  
      dependencies {
          classpath 'com.android.tools.build:gradle:1.1.1'
      }
  }
  
  project.ext {
      multiarch = false
      compileSdkVersion = Integer.parseInt(project.ANDROID_BUILD_SDK_VERSION)
      minSdkVersion = Integer.parseInt(project.ANDROID_BUILD_MIN_SDK_VERSION)
      targetSdkVersion = Integer.parseInt(project.ANDROID_BUILD_TARGET_SDK_VERSION)
      buildToolsVersion = project.ANDROID_BUILD_TOOLS_VERSION
  }
  
  dependencies {
      compile fileTree(dir: 'libs', include: '*.jar')
      compile 'com.android.support:support-v4:22.0.0'
      compile 'com.android.support:appcompat-v7:22.0.0'
      compile 'jp.co.cyberagent.android.gpuimage:gpuimage-library:1.2.3'
      compile project(':CordovaLib')
      compile project(':app-FacebookLib')
      compile files('libs/universal-image-loader-1.9.3.jar')
      compile files('libs/twitter4j-core-4.0.3.jar')
      compile files('libs/twitter4j-core-4.0.4-SNAPSHOT.jar')
      compile files('libs/Filters.jar')
  }
  
  android {
      sourceSets {
          main {
              manifest.srcFile 'AndroidManifest.xml'
              java.srcDirs = ['src']
              resources.srcDirs = ['src']
              aidl.srcDirs = ['src']
              renderscript.srcDirs = ['src']
              res.srcDirs = ['res']
              assets.srcDirs = ['assets']
          }
      }
  
      defaultConfig {
          versionCode Integer.parseInt("" + getVersionCodeFromManifest() + "0")
      }
  
      compileSdkVersion Integer.parseInt(project.ANDROID_BUILD_SDK_VERSION)
      buildToolsVersion project.ANDROID_BUILD_TOOLS_VERSION
  
      if (multiarch || System.env.BUILD_MULTIPLE_APKS) {
          productFlavors {
              armv7 {
                  versionCode defaultConfig.versionCode + 2
                  ndk {
                      abiFilters "armeabi-v7a", ""
                  }
              }
              x86 {
                  versionCode defaultConfig.versionCode + 4
                  ndk {
                      abiFilters "x86", ""
                  }
              }
              all {
                  ndk {
                      abiFilters "all", ""
                  }
              }
          }
      }
  
      compileOptions {
          sourceCompatibility JavaVersion.VERSION_1_7
          targetCompatibility JavaVersion.VERSION_1_7
      }
  
  }
  
  task wrapper(type: Wrapper) {
      gradleVersion = '1.12'
  }
  
  def getVersionCodeFromManifest() {
      def manifestFile = file(android.sourceSets.main.manifest.srcFile)
      def pattern = Pattern.compile("versionCode=\"(\\d+)\"")
      def matcher = pattern.matcher(manifestFile.getText())
      matcher.find()
      return Integer.parseInt(matcher.group(1))
  }
  </code></pre>
  
  <p>Answer:</p>
  
  <p>Needed to add </p>
  
  <pre><code>repositories {
      mavenCentral()
  }
  </code></pre>
  
  <p>To my gradle, the one I already had was in the build script, I needed it outside that scope too.</p>
  
  <p><a href="https://stackoverflow.com/a/16675271/3324388">https://stackoverflow.com/a/16675271/3324388</a></p>